digital

BASIC COMPUTER GAMES

$7.50

101 BASIC Computer Games

Digital Equipment Corporation
Maynard, Massachusetts

Additional copies of 101 BASIC Computer Games are available for $7.50 plus 50 cents postage and handling from:

Software Distribution Center
Digital Equipment Corporation
Maynard, Massachusetts 01754

Write for discount schedule on quantities over 30.

Two supplemental guides are available for use with this book. They are:

Understanding Mathematics and Logic Using BASIC Computer Games, $4.50. Grades 7-12.

Getting Started in Classroom Computing, $3.00. Grades 2-7.

1st Printing -- July 1973
2nd Printing -- April 1974
3rd Printing -- March 1975

Contents

Appendices



Preface

This is not the first collection of computer games and simulations nor will it by any means be the last. However, in many ways it is unique. It is the first collection of games all in BASIC. It is also the only collection that contains both a complete listing and a sample run of each game along with a descriptive write-up.

Educational Value of Games

Educators have widely different opinions as to the educational value of games. There tends to be agreement that games are highly motivational and frequently very addictive. Most educators agree that games generally foster learning by discovery--i.e., the player doesn't sit down at the terminal with the purpose of learning a principle of logic but after playing BAGLES three or four times he most assuredly has learned something about logic. Newton's second law is probably the furthest thing from the mind of a person sitting down to play ROCKET. However, when the player finally lands his LEM successfully on the moon, the chances are very good that he has discovered something about gravity varying inversely with the mass of the LEM and the distance from the moon.

The main objection to games as a learning tool seems to be the fact that it's largely unguided learning and potentially wasteful of computer time. Art Leuhrmann of Dartmouth joked that some computer center directors might be willing to pay to not have the book sold on campus because of the computer time that would be burned up by playing the games; however, the educational value of games can be enormous - not only in their playing but in their creation.

The majority of games submitted tend to simulate a sport, card or board game, a game of chance or something which already exists. Only a few games begin to use the logical and computational capabilities of the computer to come up with something new and truly unique. Some that do are STARES, BULCOW, ROCKET, and LIFE-2.

Certain games are, of course, more popular with game authors that others. There were no less than ten versions of NIM submitted, nine versions each of HORSES (Horse Race) and TICTAC (Tic-Tac-Toe), and eight versions of CRAPS. Other popular ones were simulations of baseball, basketball, football, blackjack, and hangman.

Families of Games

A word about the title of the book. The astute, quantitatively-oriented reader might notice that there seem to be more than 101 games in the book. In fact, there are 108 individual games; 7 are different versions of another game. There are 101 separate write-ups; thus, the title of the book.

Perhaps it is a disease of using the computer or perhaps it is just a compulsion of man that he must categorize things. The games in this book could be categorized by level of difficulty as is often the case in collections of puzzles. They could also be categorized in an educational sense, for example, those that could be used to teach logic principles, those that foster learing by discovery, those that require the user to solve an algebra problem, etc.

In the first two groups, Number or Letter Guessing and Piles of Objects, you will probably get more enjoyment if you play the games in the numbered order as there is a definite sequential nature to their difficulty. In the other fourteen categories, the games may be played in any order; one does not generally build upon another except in a few cases. In particular, you should play:

BAGLES before BULCOW
HI-Q before 1CHECK
BATTLE before SALVO
GUNNER before SUNER1
ROCKET before ROCKT2
HMRABI before KING

Equipment to Play, Computer and Otherwise

Most of the games in this book require no special knowledge, tools or equipment to play, except, of course, a BASIC-speaking computer. Four of the matrix games will probably be more enjoyable if you use a grid or quadrille paper to play. Unless you have a photographic memory, QUBIC almost certainly requires a diagram. There is a page included as Appendix B which contains some supplemental diagrams; you may wish to reproduce it if you become addicted to the games on it.

With few exceptions, the games all run in "standard" BASIC. Any exceptions are noted in the write-ups under the heading, "Computer Limitations." The major difference between various computer systems appears to be in the handling of alphabetic strings. On Digital systems a subscripted string variable, for example, A$(8) or C1$(15), refers to a variable in an array or matrix. Other BASIC compilers may not have string arrays.

On some systems, in particular, Digital's Edusystems 20, 25, and 50, strings are limited to 6 characters. Several strings may, or course, be combined in an array to permit longer than 6-letter words to be used.

Many programs use the RANDOMIZE command to start the random number generator at a random point. Some BASIC compilers do not recognize RANDOMIZE and it must be removed in order for the program to run.

Digital BASIC permits more than one statement on each program line. Statement separators on the line may be one of three characters -- / or : or ◇ .

Digital Equipment Corporation
Maynard, Massachusetts
July 1973

ROCKET

LAND AN APOLLO CAPSULE ON THE MOON

Description

ROCKET, known also as LUNAR, LEM, and APOLLO, is by far and away the single most popular computer game. It exists in versions that start you anywhere from 500 feet to 200 miles above the moon, or other planets, too. Some allow the control of directional stabilization rockets and/or the retro rocket. The three versions presented here appear to be the most popular of the many variations.

ROCKET. In this program, you set the burn rate of the retro rockets (pounds of fuel per second) every 10 seconds and attempt to achieve a soft landing on the moon. 200 lbs/sec really puts the brakes on, and 0 lbs/sec is free fall. Ignition occurs at 8 lbs/sec, so do not use burn rates between 1 and 7 lbs/sec. To make the landing more of a challenge, but more closely approximate the real Apollo LEM capsule, you should make the available fuel at the start (N) equal to 16,000 lbs, and the weight of the capsule (M) equal to 32,500 lbs in Statement 15.

Some computers object to the series expansion calculations in Statements 91 and 94 (as you near the lunar surface, these numbers get very small). If yours does, substitute the expanded form--for the expansion in Statement 91:

```
-Q*(1+Q*(1/2+Q*(1/3+Q*(1/4+Q/5))))
```

You should be able to figure the other one out yourself.

ROCKT1. In this version, you start 500 feet above the lunar surface and control the burn rate in 1-second bursts. Each unit of fuel slows your descent by 1 ft/sec. The maximum thrust of your engine is 30 ft/sec/sec.

ROCKT2. This is the most comprehensive of the three versions and permits you to control the time interval of firing, the thrust, and the attitude angle. It also allows you to work in the metric or English system of measurement. The instructions in the program dialog are very complete, so you shouldn't have any trouble.

In most versions of ROCKET, the temptation is to slow up too soon and then have no fuel left for the lower part of the journey. This, of course, is disasterous (as you will find out when you land your own capsule)!

Source

To put all the conflicting stories to rest, we can say with confidence that ROCKET was originally written in FOCAL by a Lexington High School student back in the mid 60's.

ROCKET:
Jim Storer
Lexington High School
Lexington, MA 02173

ROCKT1:
Eric Peters
Digital Equipment Corp.
Maynard, MA 01754

ROCKT2:
William Labaree II
621 Oakley Place
Alexandria, VA 22302



ROCKET PROGRAM LISTING

```
ROCKET  EDUSYSTEM 30

2 PRINT "THIS IS A COMPUTER SIMULATION OF AN APOLLO LUNAR"
3 PRINT "LANDING CAPSULE."\PRINT\PRINT
4 PRINT "THE ON-BOARD COMPUTER HAS FAILED (IT WASN'T MADE BY"
5 PRINT "DIGITAL) SO YOU HAVE TO LAND THE CAPSULE MANUALLY"
6 PRINT\PRINT "SET BURN RATE OF RETRO ROCKETS TO ANY VALUE BETWEEN"
7 PRINT "0 (FREE FALL) AND 200 (MAXIMUM BURN) POUNDS PER SECOND"
8 PRINT "SET NEW BURN RATE EVERY 10 SECONDS."\PRINT
9 PRINT "CAPSULE WEIGHT 32,500 LBS;  FUEL WEIGHT 16,500 LBS"
10 PRINT\PRINT\PRINT "GOOD LUCK!!!"
11 L=0
13 PRINT\PRINT "SEC","MI + FT","MPH","LB FUEL","BURN RATE"\PRINT
15 A=120\V=1\M=33000\N=16500\G=1E-3\Z=1.8
21 PRINT L,INT(A);INT(5280*(A-INT(A))),3600*V,M-N,\INPUT K\T=10
31 IF M-N<.001 THEN 41\IF T<.001 THEN 21\S=T\IF M>=N+S*K THEN 35
32 S=(M-N)/K
35 GOSUB 91\IF I<=0 THEN 71\IF V<=0 THEN 38\IF J<0 THEN 81
38 GOSUB 61\GOTO 31
41 PRINT "FUEL OUT AT"L"SEC"\S=(-V+SQR(V*V+2*A*G))/G\V=V+G*S\L=L+S
51 W=3600*V\PRINT"ON MOON AT"L"SEC - IMPACT VELOCITY" W "MPH"
52 IF W>1.2 THEN 53\PRINT "PERFECT LANDING! (LUCKY)"\GOTO 95
53 IF W>10 THEN 56\PRINT "GOOD LANDING (COULD BE BETTER)"\GOTO 95
56 IF W>60 THEN 58 \PRINT "CRAFT DAMAGE.... YOU'RE STRANDED HERE UNTIL"
57 PRINT "A RESCUE PARTY ARRIVES.  HOPE YOU HAVE ENOUGH OXYGEN!"\GOTO 95
58 PRINT "SORRY, BUT THERE WERE NO SURVIVORS...YOU BLEW IT!"
59 PRINT "IN FACT, YOU BLASTED A NEW LUNAR CRATER"W*.2777"FT DEEP"
60 GOTO 95
61 L=L+S\T=T-S\M=M-S*K\A=I\V=J\RETURN
71 IF S<5E-3 THEN 51\D=V+SQR(V*V+2*A*(G-Z*K/M))\S=2*A/D
73 GOSUB 91\GOSUB 61\GOTO 71
81 W=(1-M*G/(Z*K))/2\S=M*V/(Z*K*(W+SQR(W*W+V/Z)))+.05\GOSUB 91
83 IF I<=0 THEN 71\GOSUB 61\IF J>0 THEN 31\IF V>0 THEN 81\GOTO 31
91 Q=S*K/M\J=V+G*S+Z*(-Q-Q*Q/2-Q^3/3-Q^4/4-Q^5/5)
94 I=A-G*S*S/2-V*S+Z*S*(Q/2+Q^2/6+Q^3/12+Q^4/20+Q^5/30)\RETURN
95 PRINT\PRINT\PRINT\PRINT "TRY AGAIN??"\GOTO 6
99 END
```

SAMPLE RUN

```
ROCKET  EDUSYSTEM 30

THIS IS A COMPUTER SIMULATION OF AN APOLLO LUNAR
LANDING CAPSULE.


THE ON-BOARD COMPUTER HAS FAILED (IT WASN'T MADE BY
DIGITAL) SO YOU HAVE TO LAND THE CAPSULE MANUALLY

SET BURN RATE OF RETRO ROCKETS TO ANY VALUE BETWEEN
0 (FREE FALL) AND 200 (MAXIMUM BURN) POUNDS PER SECOND
SET NEW BURN RATE EVERY 10 SECONDS.

CAPSULE WEIGHT 32,500 LBS;  FUEL WEIGHT 16,500 LBS


GOOD LUCK!!!

SEC           MI + FT       MPH           LB FUEL       BURN RATE

 0             120  0        3600          16500         ?0
 10            109  5015     3636          16500         ?0
 20            99  4223      3672          16500         ?0
 30            89  2903      3708          16500         ?0
 40            79  1055      3744          16500         ?0
 50            68  3959      3780          16500         ?0
 60            58  1055      3816          16500         ?0
 70            47  2903      3852          16500         ?200
 80            37  1883      3482.87       14500         ?200
 90            28  1191      3086.7        12500         ?200
 100           20  1251      2659.65       10500         ?200
 110           13  2549      2196.94       8500          ?200
 120           8  370        1692.63       6500          ?200
 130           4  658        1139.13       4500          ?200
 140           1  4203       526.598       2500          ?100
 150           0  4042       212.242       1500          ?45
 160           0  1863       84.1831       1050          ?20
 170           0  908        45.9129       850           ?17
 180           0  438        18.107        680           ?12
 190           0  241        8.68632       560           ?11
 200           0  157        2.7691        450           ?9
 210           0  105        4.27036       360           ?9.5
 220           0  46         3.65466       265           ?9.8
 230           0  7          1.66462       167           ?9.3
ON MOON AT 233.183 SEC - IMPACT VELOCITY 1.6042 MPH
GOOD LANDING (COULD BE BETTER)


TRY AGAIN??
```

DIGITAL EQUIPMENT CORPORATION, Corporate Headquarters: Maynard, Massachusetts 01754, Telephone: (617) 897-5111

SALES AND SERVICE OFFICES

DOMESTIC — ARIZONA, Phoenix and Tucson • CALIFORNIA, Los Angeles, Monrovia, Oakland, Ridgecrest, San Diego, San Francisco (Mountain View), Santa Ana, Sunnyvale and Woodland Hills • COLORADO, Englewood • CONNECTICUT, Fairfield and Meriden • DISTRICT OF COLUMBIA, Washington (Latham, Md.) • FLORIDA, Orlando • GEORGIA, Atlanta • ILLINOIS, Chicago (Rolling Meadows) • INDIANA, Indianapolis • IOWA, Bettendorf • KENTUCKY, Louisville • LOUISIANA, Metairie (New Orleans) • MASSACHUSETTS, Marlborough and Waltham • MICHIGAN, Detroit (Farmington Hills) • MINNESOTA, Minneapolis • MISSOURI, Kansas City and St. Louis • NEW HAMPSHIRE, Manchester • NEW JERSEY, Fairfield, Metuchen and Princeton • NEW MEXICO, Albuquerque • NEW YORK, Albany, Huntington Station, Manhattan, Rochester and Syracuse • NORTH CAROLINA, Durham/Chapel Hill • OHIO, Cleveland, Columbus and Dayton • OKLAHOMA, Tulsa • OREGON, Portland • PENNSYLVANIA, Philadelphia (Bluebell) and Pittsburgh • TENNESSEE, Knoxville • TEXAS, Austin, Dallas and Houston • UTAH, Salt Lake City • WASHINGTON, Bellevue • WISCONSIN, Milwaukee (Brookfield) • INTERNATIONAL — ARGENTINA, Buenos Aires • AUSTRALIA, Adelaide, Brisbane, Canberra, Melbourne, Perth and Sydney • AUSTRIA, Vienna • BELGIUM, Brussels • BOLIVIA, La Paz • BRAZIL, Puerto Alegre, Rio de Janeiro and São Paulo • CANADA, Calgary, Halifax, Montreal Ottawa, Toronto and Vancouver • CHILE, Santiago • DENMARK, Copenhagen • FINLAND, Helsinki • FRANCE, Grenoble and Paris • GERMANY, Berlin, Cologne, Hannover, Frankfurt, Munich and Stuttgart • HONG KONG • INDIA, Bombay • INDONESIA, Djakarta • ISRAEL, Tel Aviv • ITALY, Milan and Turin • JAPAN, Osaka and Tokyo • MALAYSIA, Kuala Lumpur • MEXICO, Mexico City • NETHERLANDS, The Hague • NEW ZEALAND, Auckland • NORWAY, Oslo • PHILIPPINES, Manila • PUERTO RICO, Santurce • SINGAPORE • SPAIN, Barcelona and Madrid • SWEDEN, Stockholm • SWITZERLAND, Geneva and Zurich • TAIWAN, Taipei and Taoyuan • UNITED KINGDOM, Birmingham, Bristol, Dublin, Edinburgh, Leeds, London, Manchester and Reading • VENEZUELA, Caracas •

PRINTED IN USA ED 04873 77058/89 24 02 (P65558)